Rules for integrands of the form $\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSin}[c\,x])^n$

1. $\int\left(d+e\,x^2\right)^p\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $c^2\,d+e==0$

   1. $\int\frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2\,d+e==0$

      x: $\int\frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2\,d+e==0$

- Derivation: Piecewise constant extraction and integration by substitution

- Basis: If $c^2\,d+e==0$, then $\partial_x\frac{\sqrt{1-c^2\,x^2}}{\sqrt{d+e\,x^2}}==0$

- Basis: $\frac{F[\text{ArcSin}[c\,x]]}{\sqrt{1-c^2\,x^2}}==\frac{1}{c}\,\text{Subst}[F[x],\,x,\,\text{ArcSin}[c\,x]]\,\partial_x\text{ArcSin}[c\,x]$

- Note: When $n==1$, this rule would result in a slightly less compact antiderivative since $\int(a+b\,x)^n\,dx$ returns a sum.

- Rule: If $c^2\,d+e==0$, then

$$\int\frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx\,\rightarrow\,\frac{\sqrt{1-c^2\,x^2}}{c\,\sqrt{d+e\,x^2}}\,\text{Subst}\left[\int(a+b\,x)^n\,dx,\,x,\,\text{ArcSin}[c\,x]\right]$$

- Program code:

```
(* Int[(a_.+b_.*ArcSin[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/c*Simp[Sqrt[1-c^2*x^2]/Sqrt[d+e*x^2]]*Subst[Int[(a+b*x)^n,x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] *)
```

```
(* Int[(a_.+b_.*ArcCos[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  -1/c*Simp[Sqrt[1-c^2*x^2]/Sqrt[d+e*x^2]]*Subst[Int[(a+b*x)^n,x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] *)
```

1: $\int \frac{1}{\sqrt{d+e x^2}\,(a+b \operatorname{ArcSin}[c x])}\,dx$ when $c^2 d+e==0$

- Derivation: Piecewise constant extraction and integration by substitution

- Basis: If $c^2 d+e==0$, then $\partial_x \frac{\sqrt{1-c^2 x^2}}{\sqrt{d+e x^2}}==0$

- Rule: If $c^2 d+e==0$, then

$$\int \frac{1}{\sqrt{d+e x^2}\,(a+b \operatorname{ArcSin}[c x])}\,dx \longrightarrow \frac{\sqrt{1-c^2 x^2}}{b c \sqrt{d+e x^2}} \operatorname{Log}[a+b \operatorname{ArcSin}[c x]]$$

- Program code:

```
Int[1/(Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSin[c_.*x_])),x_Symbol] :=
  1/(b*c)*Simp[Sqrt[1-c^2*x^2]/Sqrt[d+e*x^2]]*Log[a+b*ArcSin[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0]
```

```
Int[1/(Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcCos[c_.*x_])),x_Symbol] :=
  -1/(b*c)*Simp[Sqrt[1-c^2*x^2]/Sqrt[d+e*x^2]]*Log[a+b*ArcCos[c*x]]/(b*c*Sqrt[d]) /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0]
```

2: $\int \frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2 d+e=0\land n\neq -1$

–

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $c^2 d+e=0$, then $\partial_x \frac{\sqrt{1-c^2 x^2}}{\sqrt{d+e x^2}}=0$

Rule: If $c^2 d+e=0\land n\neq -1$, then

$$\int \frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \longrightarrow \frac{\sqrt{1-c^2 x^2}}{b\,c\,(n+1)\,\sqrt{d+e\,x^2}}\,(a+b\,\text{ArcSin}[c\,x])^{n+1}$$

Program code:

```
Int[(a_.+b_.*ArcSin[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/(b*c*(n+1))*Simp[Sqrt[1-c^2*x^2]/Sqrt[d+e*x^2]]*(a+b*ArcSin[c*x])^(n+1) /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] && NeQ[n,-1]
```

```
Int[(a_.+b_.*ArcCos[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  -1/(b*c*(n+1))*Simp[Sqrt[1-c^2*x^2]/Sqrt[d+e*x^2]]*(a+b*ArcCos[c*x])^(n+1) /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] && NeQ[n,-1]
```

2. $\int (d+e\,x^2)^p\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $c^2 d+e=0\land n>0$

1: $\int (d+e\,x^2)^p\,(a+b\,\text{ArcSin}[c\,x])\,dx$ when $c^2 d+e=0\land p\in\mathbb{Z}^+$

Derivation: Integration by parts

–

Rule: If $c^2 d+e=0\land p\in\mathbb{Z}^+$, let $u\to\int (d+e\,x^2)^p\,dx$, then

$$\int \left(d+e\,x^2\right)^p \left(a+b\,\text{ArcSin}[c\,x]\right) dx \longrightarrow u \left(a+b\,\text{ArcSin}[c\,x]\right) - b\,c \int \frac{u}{\sqrt{1-c^2\,x^2}}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcSin[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[p,0]
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcCos[c*x],u,x] + b*c*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[p,0]
```

2. $\int (d+e x^2)^p (a+b\,\text{ArcSin}[c\,x])^n \,dx$ when $c^2 d+e==0 \land n>0 \land p>0$

1: $\int \sqrt{d+e x^2}\, (a+b\,\text{ArcSin}[c\,x])^n \,dx$ when $c^2 d+e==0 \land n>0$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $c^2 d+e==0 \land n>0$, then

$$\int \sqrt{d+e x^2}\, (a+b\,\text{ArcSin}[c\,x])^n \,dx \longrightarrow$$

$$\frac{x\sqrt{d+e x^2}\,(a+b\,\text{ArcSin}[c\,x])^n}{2}-\frac{b\,c\,n\sqrt{d+e x^2}}{2\sqrt{1-c^2 x^2}}\int x\,(a+b\,\text{ArcSin}[c\,x])^{n-1}\,dx+\frac{\sqrt{d+e x^2}}{2\sqrt{1-c^2 x^2}}\int\frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{1-c^2 x^2}}\,dx$$

–

Program code:

```
Int[Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  x*Sqrt[d+e*x^2]*(a+b*ArcSin[c*x])^n/2 -
  b*c*n/2*Simp[Sqrt[d+e*x^2]/Sqrt[1-c^2*x^2]]*Int[x*(a+b*ArcSin[c*x])^(n-1),x] +
  1/2*Simp[Sqrt[d+e*x^2]/Sqrt[1-c^2*x^2]]*Int[(a+b*ArcSin[c*x])^n/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0]
```

```
Int[Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  x*Sqrt[d+e*x^2]*(a+b*ArcCos[c*x])^n/2 +
  b*c*n/2*Simp[Sqrt[d+e*x^2]/Sqrt[1-c^2*x^2]]*Int[x*(a+b*ArcCos[c*x])^(n-1),x] +
  1/2*Simp[Sqrt[d+e*x^2]/Sqrt[1-c^2*x^2]]*Int[(a+b*ArcCos[c*x])^n/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0]
```

2: $\int (d+e x^2)^p (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \land n>0 \land p>0$

Derivation: Inverted integration by parts

Rule: If $c^2 d+e=0 \land n>0 \land p>0$, then

$$\int (d+e x^2)^p (a+b \operatorname{ArcSin}[c x])^n \, dx \longrightarrow$$

$$\frac{x (d+e x^2)^p (a+b \operatorname{ArcSin}[c x])^n}{2 p+1}+$$

$$\frac{2 d p}{2 p+1} \int (d+e x^2)^{p-1} (a+b \operatorname{ArcSin}[c x])^n \, dx-\frac{b c n (d+e x^2)^p}{(2 p+1) (1-c^2 x^2)^p} \int x (1-c^2 x^2)^{p-\frac{1}{2}} (a+b \operatorname{ArcSin}[c x])^{n-1} \, dx$$

–

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  x*(d+e*x^2)^p*(a+b*ArcSin[c*x])^n/(2*p+1) +
  2*d*p/(2*p+1)*Int[(d+e*x^2)^(p-1)*(a+b*ArcSin[c*x])^n,x] -
  b*c*n/(2*p+1)*Simp[(d+e*x^2)^p/(1-c^2*x^2)^p]*Int[x*(1-c^2*x^2)^(p-1/2)*(a+b*ArcSin[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && GtQ[p,0]
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  x*(d+e*x^2)^p*(a+b*ArcCos[c*x])^n/(2*p+1) +
  2*d*p/(2*p+1)*Int[(d+e*x^2)^(p-1)*(a+b*ArcCos[c*x])^n,x] +
  b*c*n/(2*p+1)*Simp[(d+e*x^2)^p/(1-c^2*x^2)^p]*Int[x*(1-c^2*x^2)^(p-1/2)*(a+b*ArcCos[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && GtQ[p,0]
```

3. $\int (d + e x^2)^p (a + b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d + e == 0 \land n > 0 \land p < -1$

1: $\int \frac{(a + b \operatorname{ArcSin}[c x])^n}{(d + e x^2)^{3/2}} \, dx$ when $c^2 d + e == 0 \land n > 0$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{1}{(d+e x^2)^{3/2}} == \partial_x \frac{x}{d \sqrt{d+e x^2}}$

Basis: If $c^2 d + e == 0$, then $\partial_x \frac{\sqrt{1-c^2 x^2}}{\sqrt{d+e x^2}} == 0$

Rule: If $c^2 d + e == 0 \land n > 0$, then

$$\int \frac{(a + b \operatorname{ArcSin}[c x])^n}{(d + e x^2)^{3/2}} \, dx \rightarrow \frac{x (a + b \operatorname{ArcSin}[c x])^n}{d \sqrt{d + e x^2}} - \frac{b c n \sqrt{1 - c^2 x^2}}{d \sqrt{d + e x^2}} \int \frac{x (a + b \operatorname{ArcSin}[c x])^{n-1}}{1 - c^2 x^2} \, dx$$

Program code:

```
Int[(a_.+b_.*ArcSin[c_.*x_])^n_./(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  x*(a+b*ArcSin[c*x])^n/(d*Sqrt[d+e*x^2]) -
  b*c*n/d*Simp[Sqrt[1-c^2*x^2]/Sqrt[d+e*x^2]]*Int[x*(a+b*ArcSin[c*x])^(n-1)/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0]
```

```
Int[(a_.+b_.*ArcCos[c_.*x_])^n_./(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  x*(a+b*ArcCos[c*x])^n/(d*Sqrt[d+e*x^2]) +
  b*c*n/d*Simp[Sqrt[1-c^2*x^2]/Sqrt[d+e*x^2]]*Int[x*(a+b*ArcCos[c*x])^(n-1)/(1-c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0]
```

2: $\int (d + e x^2)^p (a + b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d + e == 0 \land n > 0 \land p < -1 \land p \neq -\frac{3}{2}$

Rule: If $c^2 d + e == 0 \land n > 0 \land p < -1 \land p \neq -\frac{3}{2}$, then

$$\int \left(d+e x^2\right)^p \left(a+b\,\text{ArcSin}[c\,x]\right)^n dx \longrightarrow$$

$$-\frac{x\left(d+e x^2\right)^{p+1}\left(a+b\,\text{ArcSin}[c\,x]\right)^n}{2\,d\,(p+1)}+$$

$$\frac{2p+3}{2\,d\,(p+1)}\int\left(d+e x^2\right)^{p+1}\left(a+b\,\text{ArcSin}[c\,x]\right)^n dx+\frac{b\,c\,n\left(d+e x^2\right)^p}{2\,(p+1)\left(1-c^2 x^2\right)^p}\int x\left(1-c^2 x^2\right)^{p+\frac{1}{2}}\left(a+b\,\text{ArcSin}[c\,x]\right)^{n-1} dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  -x*(d+e*x^2)^(p+1)*(a+b*ArcSin[c*x])^n/(2*d*(p+1)) +
  (2*p+3)/(2*d*(p+1))*Int[(d+e*x^2)^(p+1)*(a+b*ArcSin[c*x])^n,x] +
  b*c*n/(2*(p+1))*Simp[(d+e*x^2)^p/(1-c^2*x^2)^p]*Int[x*(1-c^2*x^2)^(p+1/2)*(a+b*ArcSin[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2]
```

```
Int[(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  -x*(d+e*x^2)^(p+1)*(a+b*ArcCos[c*x])^n/(2*d*(p+1)) +
  (2*p+3)/(2*d*(p+1))*Int[(d+e*x^2)^(p+1)*(a+b*ArcCos[c*x])^n,x] -
  b*c*n/(2*(p+1))*Simp[(d+e*x^2)^p/(1-c^2*x^2)^p]*Int[x*(1-c^2*x^2)^(p+1/2)*(a+b*ArcCos[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[p,-1] && NeQ[p,-3/2]
```

4: $\int \frac{(a+b\,\text{ArcSin}[c\,x])^n}{d+e\,x^2}\,dx$ when $c^2 d+e==0 \land n\in\mathbb{Z}^+$

- Derivation: Integration by substitution

- Basis: If $c^2 d+e==0$, then $\frac{1}{d+e\,x^2}==\frac{1}{c\,d}\text{Sec}[\text{ArcSin}[c\,x]]\,\partial_x\text{ArcSin}[c\,x]$

- Note: If $n\in\mathbb{Z}^+$, then $(a+b\,x)^n\,\text{Sec}[x]$ is integrable in closed-form.

- Rule: If $c^2 d+e==0 \land n\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\text{ArcSin}[c\,x])^n}{d+e\,x^2}\,dx \longrightarrow \frac{1}{c\,d}\,\text{Subst}\left[\int (a+b\,x)^n\,\text{Sec}[x]\,dx,\ x,\ \text{ArcSin}[c\,x]\right]$$

- Program code:

```
Int[(a_.+b_.*ArcSin[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  1/(c*d)*Subst[Int[(a+b*x)^n*Sec[x],x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCos[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  -1/(c*d)*Subst[Int[(a+b*x)^n*Csc[x],x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[n,0]
```

3. $\int (d+e x^2)^p (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \land n<-1$

1: $\int (d+e x^2)^p (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \land n<-1 \land (p \in \mathbb{Z} \lor d>0)$

Derivation: Integration by parts

Basis: $\frac{(a+b \operatorname{ArcSin}[c x])^n}{\sqrt{1-c^2 x^2}} = \partial_x \frac{(a+b \operatorname{ArcSin}[c x])^{n+1}}{b c (n+1)}$

Rule: If $c^2 d+e=0 \land n<-1 \land (p \in \mathbb{Z} \lor d>0)$, then

$$\int (d+e x^2)^p (a+b \operatorname{ArcSin}[c x])^n \, dx \to$$

$$\frac{d^p \left(1-c^2 x^2\right)^{p+\frac{1}{2}} (a+b \operatorname{ArcSin}[c x])^{n+1}}{b c (n+1)}+\frac{c d^p (2 p+1)}{b (n+1)} \int x \left(1-c^2 x^2\right)^{p-\frac{1}{2}} (a+b \operatorname{ArcSin}[c x])^{n+1} \, dx$$

Program code:

```
(* Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  d^p*(1-c^2*x^2)^(p+1/2)*(a+b*ArcSin[c*x])^(n+1)/(b*c*(n+1)) +
  c*d^p*(2*p+1)/(b*(n+1))*Int[x*(1-c^2*x^2)^(p-1/2)*(a+b*ArcSin[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1] && (IntegerQ[p] || GtQ[d,0]) *)
```

```
(* Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -d^p*(1-c^2*x^2)^(p+1/2)*(a+b*ArcCos[c*x])^(n+1)/(b*c*(n+1)) -
  c*d^p*(2*p+1)/(b*(n+1))*Int[x*(1-c^2*x^2)^(p-1/2)*(a+b*ArcCos[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1] && (IntegerQ[p] || GtQ[d,0]) *)
```

2: $\int (d+e x^2)^p (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d+e=0 \land n<-1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b\,\text{ArcSin}[c\,x])^n}{\sqrt{1-c^2x^2}} == \partial_x \frac{(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Basis: If $c^2 d + e == 0$, then $\partial_x \frac{(d+e\,x^2)^p}{(1-c^2x^2)^p} == 0$

Rule: If $c^2 d + e == 0 \land n < -1$, then

$$\int (d+e\,x^2)^p\,(a+b\,\text{ArcSin}[c\,x])^n\,dx \longrightarrow$$

$$\frac{\sqrt{1-c^2x^2}\,(d+e\,x^2)^p\,(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)} + \frac{c\,(2p+1)}{b\,(n+1)}\int \frac{x\,(d+e\,x^2)^p\,(a+b\,\text{ArcSin}[c\,x])^{n+1}}{\sqrt{1-c^2x^2}}\,dx \longrightarrow$$

$$\frac{\sqrt{1-c^2x^2}\,(d+e\,x^2)^p\,(a+b\,\text{ArcSin}[c\,x])^{n+1}}{b\,c\,(n+1)} + \frac{c\,(2p+1)\,(d+e\,x^2)^p}{b\,(n+1)\,(1-c^2x^2)^p}\int x\,(1-c^2x^2)^{p-\frac{1}{2}}\,(a+b\,\text{ArcSin}[c\,x])^{n+1}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  Sqrt[1-c^2*x^2]*(d+e*x^2)^p*(a+b*ArcSin[c*x])^(n+1)/(b*c*(n+1)) +
  c*(2*p+1)/(b*(n+1))*Simp[(d+e*x^2)^p/(1-c^2*x^2)^p]*Int[x*(1-c^2*x^2)^(p-1/2)*(a+b*ArcSin[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1]
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -Sqrt[1-c^2*x^2]*(d+e*x^2)^p*(a+b*ArcCos[c*x])^(n+1)/(b*c*(n+1)) -
  c*(2*p+1)/(b*(n+1))*Simp[(d+e*x^2)^p/(1-c^2*x^2)^p]*Int[x*(1-c^2*x^2)^(p-1/2)*(a+b*ArcCos[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1]
```

4: $\int (d + e x^2)^p (a + b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d + e == 0 \land 2 p \in \mathbb{Z}^+$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $c^2 d + e == 0$, then $\partial_x \frac{(d+e x^2)^p}{(1-c^2 x^2)^p} == 0$

Basis: $(1 - c^2 x^2)^p == \frac{1}{b c} \operatorname{Subst}\left[\operatorname{Cos}\left[-\frac{a}{b} + \frac{x}{b}\right]^{2 p+1}, x, a + b \operatorname{ArcSin}[c x]\right] \partial_x (a + b \operatorname{ArcSin}[c x])$

Note: If $2 p \in \mathbb{Z}^+$, then $x^n \operatorname{Cos}\left[\frac{a}{b} - \frac{x}{b}\right]^{2 p+1}$ is integrable in closed-form.

Rule: If $c^2 d + e == 0 \land 2 p \in \mathbb{Z}^+$, then

$$\int (d + e x^2)^p (a + b \operatorname{ArcSin}[c x])^n \, dx \rightarrow \frac{(d + e x^2)^p}{(1 - c^2 x^2)^p} \int (1 - c^2 x^2)^p (a + b \operatorname{ArcSin}[c x])^n \, dx$$

$$\rightarrow \frac{(d + e x^2)^p}{b c (1 - c^2 x^2)^p} \operatorname{Subst}\left[\int x^n \operatorname{Cos}\left[-\frac{a}{b} + \frac{x}{b}\right]^{2 p+1} dx, x, a + b \operatorname{ArcSin}[c x]\right]$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  1/(b*c)*Simp[(d+e*x^2)^p/(1-c^2*x^2)^p]*Subst[Int[x^n*Cos[-a/b+x/b]^(2*p+1),x],x,a+b*ArcSin[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] && IGtQ[2*p,0]
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  -1/(b*c)*Simp[(d+e*x^2)^p/(1-c^2*x^2)^p]*Subst[Int[x^n*Sin[-a/b+x/b]^(2*p+1),x],x,a+b*ArcCos[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] && IGtQ[2*p,0]
```

2. $\int (d+e x^2)^p (a+b \operatorname{ArcSin}[c x])^n \, dx$ when $c^2 d+e \neq 0$

1: $\int (d+e x^2)^p (a+b \operatorname{ArcSin}[c x]) \, dx$ when $c^2 d+e \neq 0 \land \left(p \in \mathbb{Z}^+ \lor p+\frac{1}{2} \in \mathbb{Z}^-\right)$

Derivation: Integration by parts

Rule: If $c^2 d+e \neq 0 \land \left(p \in \mathbb{Z}^+ \lor p+\frac{1}{2} \in \mathbb{Z}^-\right)$, let $u \to \int (d+e x^2)^p \, dx$, then

$$\int (d+e x^2)^p (a+b \operatorname{ArcSin}[c x]) \, dx \longrightarrow u (a+b \operatorname{ArcSin}[c x]) - b c \int \frac{u}{\sqrt{1-c^2 x^2}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcSin[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d+e,0] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^p,x]},
  Dist[a+b*ArcCos[c*x],u,x] + b*c*Int[SimplifyIntegrand[u/Sqrt[1-c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c^2*d+e,0] && (IGtQ[p,0] || ILtQ[p+1/2,0])
```

2: $\int (d+e x^2)^p (a+b \text{ArcSin}[c x])^n \, dx$ when $c^2 d+e \neq 0 \land p \in \mathbb{Z} \land (p>0 \lor n \in \mathbb{Z}^+)$

Derivation: Algebraic expansion

Rule: If $c^2 d+e \neq 0 \land p \in \mathbb{Z} \land (p>0 \lor n \in \mathbb{Z}^+)$, then

$$\int (d+e x^2)^p (a+b \text{ArcSin}[c x])^n \, dx \to \int (a+b \text{ArcSin}[c x])^n \, \text{ExpandIntegrand}\left[(d+e x^2)^p, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSin[c*x])^n,(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && NeQ[c^2*d+e,0] && IntegerQ[p] && (GtQ[p,0] || IGtQ[n,0])
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCos[c*x])^n,(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,n},x] && NeQ[c^2*d+e,0] && IntegerQ[p] && (GtQ[p,0] || IGtQ[n,0])
```

U: $\int (d+e x^2)^p (a+b \text{ArcSin}[c x])^n \, dx$

Rule:

$$\int (d+e x^2)^p (a+b \text{ArcSin}[c x])^n \, dx \to \int (d+e x^2)^p (a+b \text{ArcSin}[c x])^n \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d+e*x^2)^p*(a+b*ArcSin[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,n,p},x]
```

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(d+e*x^2)^p*(a+b*ArcCos[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,n,p},x]
```

Rules for integrands of the form $(d+e x)^p (f+g x)^q (a+b \text{ArcSin}[c x])^n$

1: $\int (d+e x)^p (f+g x)^q (a+b \text{ArcSin}[c x])^n \, dx$ when $e f+d g == 0 \land c^2 d^2-e^2 == 0 \land (p \mid q) \in \mathbb{Z}+\frac{1}{2} \land p-q \geq 0 \land d>0 \land \frac{g}{e}<0$

Derivation: Algebraic expansion

Basis: If $e f+d g == 0 \land c^2 d^2-e^2 == 0 \land d>0 \land \frac{g}{e}<0$, then
$(d+e x)^p (f+g x)^q == \left(-\frac{d^2 g}{e}\right)^q (d+e x)^{p-q} \left(1-c^2 x^2\right)^q$

Rule: If $e f+d g == 0 \land c^2 d^2-e^2 == 0 \land (p \mid q) \in \mathbb{Z}+\frac{1}{2} \land p-q \geq 0 \land d>0 \land \frac{g}{e}<0$, then

$$\int (d+e x)^p (f+g x)^q (a+b \text{ArcSin}[c x])^n \, dx \longrightarrow \left(-\frac{d^2 g}{e}\right)^q \int (d+e x)^{p-q} \left(1-c^2 x^2\right)^q (a+b \text{ArcSin}[c x])^n \, dx$$

Program code:

```
Int[(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  (-d^2*g/e)^q*Int[(d+e*x)^(p-q)*(1-c^2*x^2)^q*(a+b*ArcSin[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2-e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0] && GtQ[d,0] && LtQ[g/e,0]
```

```
Int[(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  (-d^2*g/e)^q*Int[(d+e*x)^(p-q)*(1-c^2*x^2)^q*(a+b*ArcCos[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2-e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0] && GtQ[d,0] && LtQ[g/e,0]
```

2: $\int (d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$ when $e\,f+d\,g==0\land c^2\,d^2-e^2==0\land (p\mid q)\in\mathbb{Z}+\frac{1}{2}\land p-q\ge 0\land\neg\left(d>0\land\frac{g}{e}<0\right)$

Derivation: Piecewise constant extraction

Basis: If $e\,f+d\,g==0\land c^2\,d^2-e^2==0$, then $\partial_x\frac{(d+e\,x)^q\,(f+g\,x)^q}{\left(1-c^2\,x^2\right)^q}==0$

Rule: If $e\,f+d\,g==0\land c^2\,d^2-e^2==0\land (p\mid q)\in\mathbb{Z}+\frac{1}{2}\land p-q\ge 0\land\neg\left(d>0\land\frac{g}{e}<0\right)$, then

$$\int (d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSin}[c\,x])^n\,dx\;\rightarrow\;\frac{(d+e\,x)^q\,(f+g\,x)^q}{\left(1-c^2\,x^2\right)^q}\int (d+e\,x)^{p-q}\,\left(1-c^2\,x^2\right)^q\,(a+b\,\text{ArcSin}[c\,x])^n\,dx$$

Program code:

```
Int[(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSin[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x)^q*(f+g*x)^q/(1-c^2*x^2)^q*Int[(d+e*x)^(p-q)*(1-c^2*x^2)^q*(a+b*ArcSin[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2-e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0]
```

```
Int[(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcCos[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x)^q*(f+g*x)^q/(1-c^2*x^2)^q*Int[(d+e*x)^(p-q)*(1-c^2*x^2)^q*(a+b*ArcCos[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2-e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0]
```